
TTCN‑3 supports ordered structured types known as record. The elements fields of a record type may be any of the basic types or user-defined data types (such as other records, sets or arrays). The values of a record shall be compatible with the types of the record fields. The element field identifiers are local to the record and shall be unique within the record (but do not have to be globally unique). 
EXAMPLE 1:
	type record MyRecordType
	{	
		integer 			field1,
		MyOtherRecordType 	field2 optional,
		charstring 			field3
	}

	type record MyOtherRecordType
	{	
		bitstring 	field1,
		boolean 	field2
	}

Records may be defined with no fields, i.e. as empty records.
EXAMPLE 2:
	type record MyEmptyRecord {}

A record value is assigned on an individual element field basis. The order of field values in the value list notation shall be the same as the order of fields in the related type definition.
EXAMPLE 3:
	var integer v_myIntegerValue := 1;

	const MyOtherRecordType c_myOtherRecordValue:= 
	{	
		field1 := '11001'B,
		field2 := true 
	}

	var MyRecordType v_myRecordValue := 
	{	
		field1 := v_myIntegerValue, 
		field2 := c_myOtherRecordValue,
		field3 := "A string"
	}

The same value specified with a value list.
EXAMPLE 4:
	v_myRecordValue:= {v_myIntegerValue, {'11001'B, true}, "A string"};

When the assignment notation is used for record‑s, fields wished to be changed shall be identified explicitly and a value, the not used symbol "-" or the omit keyword can be associated with them. The omit keyword shall only be used for optional fields. Its result is that the given field is not present in the given value.
Mandatory fields, not explicitly referred to in the notation or explicitely unspecified using the not used symbol "-", shall remain unchanged. In particular, when specifying partial values (i.e. setting the value of only a subset of the fields) using the assignment notation, at initialization, only the fields to be assigned values shall be specified. Fields not mentioned are implicitly left uninitialized. When re-assigning a previously initialized value, using the not used symbol or just skipping a field in an assignment notation, will cause that field to remain unchanged. Even when specifying partial values each field shall not appear more than once.
NOTE:	Please note the difference between omitted and uninitialized fields. Omitted optional fields are not present in the record or set value intentionally, i.e. the field is initialized and it does not prevent the whole record or set from being completely initialized.
When the assignment notation is used in a scope, where the optional attribute is implicitly or explicitly set to "explicit omit", fields, not explicitly referred to in the notation, shall remain unchanged. In particular, when specifying partial values (i.e. setting the value of only a subset of the fields) using the assignment notation, for example, at initialization, only the fields or elements to be assigned values shall be specified. Fields or elements not mentioned are implicitly left uninitializedWhen re-assigning a previously initialized value, using the not used symbol or just skipping a field or element in an assignment notation, will cause that field or element to remain unchanged.
EXAMPLE 5:
	type record MyRecordType
	{	
		bitstring 		field1,
		boolean			field2 optional,
		charstring 		field3
	}

	var MyRecordType v_myVariable := 
	{
		field1 := '111'B,
		field2 := false,
		field3 := -
	}

	v_myVariable := { '10111'B, -, - };
	// after this, v_myVariable contains:
	//	{ '10111'B, false /* unchanged */, <undefined> /* unchanged */ }

	v_myVariable :=
	{
		field2 := true
	}
	// after this, v_myVariable contains:
	//	{ '10111'B /* unchanged */, true, <undefined> /* unchanged */ }

	v_myVariable :=
	{
		field1 := -,
		field2 := false,
		field3 := -
	}
	// after this, v_myVariable contains:
	//	{ '10111'B /* unchanged */, false, <undefined> /* unchanged */}

When the assignment notation is used in a scope, where the optional attribute is implicitly or explicitly set to "explicit omit", optional and mandatory fields, not directly referred to in the notation shall remain unchanged.
When optional fields of variables are not assigned explicitly, they are uninitialized (i.e the optional attribute shall not have any effect on variables as described in section 27.7 restriction a ).
When the assignment notation is used in a scope, where the optional attribute is set to "implicit omit", optional fields, not directly referred to in the notation, shall implicitly be set to omit, while mandatory fields shall remain unchanged (see also clause 27.7).
EXAMPLE 6:
	type record MyRecordType
	{	
		bitstring 		field1,
		boolean			field2 optional,
		charstring 		field3
	}

	const MyRecordType c_myConst1 := 
	{
		field1 := '111'B,
		field3 := “A string”
	} //	{ '10111'B, <undefined>, “A string”}

	const MyRecordType c_myConst2 := 
	{
		field1 := '111'B,
		field3 := “A string”
	} with { optional "implicit omit" }
	//	{ '10111'B, omit /* because of the optional attribute */, “A string”}

When using value list notation in a scope where the optional attribute is implicitly or explicitly set to "explicit omit", all remaining fields at the end of the type definition, missing from the value list notation,are left unchanged.
When the value list notation is used in a scope, where the optional attribute is implicitly or explicitly set to "explicit omit, already initialized fields or elements left without an associated component in a value list notation (i.e. at the end of a value ) are becoming uninitialized. In this way, a value with initialized fields or elements can be made empty by using an empty pair of curly brackets ("{}").
When using value list notation in a scope where the optional attribute is set to "implicit omit", optional fields wished to be omitted by the implicit mechanism, but followed by fields to which a value or template is assigned explicitly, shall be skipped by using the not used symbol "-". When all remaining fields at the end of the type definition are optional and they are wished to be omitted by the implicit mechanism, either the not used symbol "-" can be used for some or all of them or they can simply be left out from the notation.
	
EXAMPLE 76:
	type record R {
		integer f1,
		integer f2 optional,
		integer f3,
		integer f4 optional,
		integer f5 optional
	}

	constvar  R vc_x := { 1, -, 2 } with { optional "implicit omit" }
	// after the assignment vc_x contains { 1, omit, 2, omit, omit }
	var const R vc_x2 := { 1, 2, 3, - } with { optional "implicit omit" }
	//  after the assignment vc_x2 contains { 1, 2, <undefined>3, omit, omit } 

When using direct assignment notation in a scope where the optional attribute is set to "implicit omit", the uninitialized optional fields in the referenced value, shall implicitly be set to omit after the assignment in the new value, while mandatory fields shall remain unchanged (see also clause 27.7)
EXAMPLE 8:
const R c_x3 := { 1, -, 2 } 
// after the assignment c_x3 contains { 1, <undefined>, 2, <undefined>, <undefined>}
const R c_x4 := c_x3 with { optional "implicit omit" }
//  after the assignment c_x4 contains { 1, omit, 2, omit, omit }

Normally, a template specifies a set of base or default values or matching symbols for each and every field defined in the appropriate type or signature definition. In cases where small changes are needed to specify a new template, it is possible to specify a modified template. A modified template specifies modifications to particular fields of the original template, either directly or indirectly. As well as creating explicitly named modified templates, TTCN‑3 allows the definition of in-line modified templates.
Syntactical Structure
Global or local modified template:
template [restriction] [ @fuzzy ] Type TemplateIdentifier  ["(" TemplateFormalParList ")"] 
modifies TemplateRef ":=" TemplateBody

NOTE 1:	The optional restriction part is covered by clause 15.8.
In-line modified template:
[ Type ":" ] modifies TemplateRefWithParList ":=" TemplateBody

Semantic Description
The modifies keyword denotes the parent template from which the new, or modified template shall be derived. This parent template may be either an original template or a modified template.
The modifications occur in a linked fashion, eventually tracing back to the original template.
a)	In case of templates, template fields or list elements of simple types, union and enumerated types, the matching mechanism specified in the modified template is simply replacing its corresponding content in its parent.
b)	For templates, template fields and elements of record and set types, if a record or set field and its corresponding matching mechanism is specified in the modified template, then the specified matching mechanism replaces the one specified in the corresponding field of the parent template. If a record or set field or its corresponding matching mechanism is – implicitly or explicitly by using the not used symbol "‑" - left unspecified in the modified template, then the matching mechanism in the corresponding field of the parent template shall be used. When the field to be modified is nested within a template field which is a structured field itself, no other field of the structured field is changed apart from the explicitly denoted one(s).
c)	For templates, template fields and elements of record of and set of types, the above rules specified for records and sets apply with the following deviations:
if the value list notation is used, only the number of elements listed in the modified template is inherited from the parent (i.e. the list is truncated at the last element of the list notation in the modified template);
when individual values of a modified template or a modified template field of record of or set of type wished to be changed, and only in these cases, the index assignment notation may also be used, where the left hand side of the assignment is the index of the element to be altered.
In case of record of and set of types first apply rule (c) to the complete structure (e.g. truncation) than apply further rules for the remaining individual type structure elements (see example 3).
Formal value or template parameters of modified templates inherit the default value or respectively template of the corresponding parameter of their parent templates only, if this is denoted by the dash (don't change) symbol at the place of the parameters' default value or respectively template.
Modified templates may also be restricted. Template restrictions are specified in clause 15.8.
A modified template may also be declared fuzzy using the @fuzzy modifier. 
NOTE 2:	If a fuzzy modified template modifies a non-fuzzy unparameterized template, the inherited fields before modification will be the same for every evaluation of the fuzzy template.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	A modified template shall not refer to itself, either directly or indirectly, i.e. recursive derivation is not allowed.
b)	If a base template has a formal parameter list, the following rules apply to all modified templates derived from that base template, whether or not they are derived in one or several modification steps:
1)	the derived template shall not omit parameters and change types or names of parameters defined at any of the modification steps between the base template and the actual modified template;
2)	a template parameter restriction of a derived template specified at any of the modification steps between the base template and the actual modified template can be changed to a stricter one (see clause 15.8);
3)	a derived template can have additional (appended) parameters if wished;
4)	if the dash (don't change) symbol is used at the place of a default value or default template, the corresponding parameter of the parent template shall have a valid default value or default template, either assigned directly or inherited. If not, this shall cause an error.
c)	Restrictions on referencing elements of templates or template fields are described in clause 15.6: for modified templates the rules for the left hand side of assignments apply.
d)	Limitations on template restrictions described in clause 15.8 shall apply.
Examples
EXAMPLE 1:	Modifying record templates (non-embedded case)
	// Modifying records
	type record MyRecordType
	{
		integer field1 optional,
		charstring field2,
		boolean field3
	}
	template MyRecordType m_myRecTemplate1 :=
	{
		field1 := 123,
		field2 := "A string",
		field3 := true
	}
	// then writing 
	template MyRecordType m_myRecTemplate2 modifies m_myRecTemplate1 :=
	{
		field1 := omit,					// field1 is optional but present in m_myTemplate1
		field2 := "A modified string"
										// field3 is unchanged
	}
	// is the same as writing 
	// template MyRecordType m_myRecTemplate2 :=
	// {
	//	  field1 := omit,
	//	  field2 := "A modified string",
	//	  field3 := true
	// }

	template MyRecordType m_myRecTemplate3 modifies m_myRecTemplate1 := {omit, "A modified string"}
	//field3 is implicitly left unchanged;
	//m_myRecTemplate3 has the same content as m_myRecTemplate2

	template MyRecordType m_myRecTemplate4 modifies m_myRecTemplate1 := {omit,"A modified string",-}
	//field3 is explicitly left unchanged;
	//m_myRecTemplate4 has the same content as m_myRecTemplate2 and m_myRecTemplate3

EXAMPLE 2:	Modifying record of templates (non-embedded case)
	type record of integer MyRecordOfType;

	template MyRecordOfType m_myBaseTemplate := { 0, 1, 2, 3, 4, 5, 6, 7, 8, 9 };

	template MyRecordOfType m_myRecOfTemplate1 modifies m_myBaseTemplate := 
                                                                 { -, -, 3, 2, -, -, -, -, -, - };
	// m_myRecOfTemplate1 contains { 0, 1, 3, 2, 4, 5, 6, 7, 8, 9 }

	template MyRecordOfType m_myRecOfTemplate2 modifies m_myBaseTemplate := { -, -, 3, 2 };
	// m_myRecOfTemplate2 replaces m_myBaseTemplate with: { 0, 1, 3, 2 };
	// elements 5 to 10 of m_myBaseTemplate are truncated

	template MyRecordOfType m_myRecOfTemplate3 modifies m_myBaseTemplate := { [2] := 3, [3] := 2 };
	// m_myRecOfTemplate3 has the same content as m_myMod1Template: { 0, 1, 3, 2, 4, 5, 6, 7, 8, 9 }

EXAMPLE 3:	Modifying embedded record and record of templates
	//Modifying a record embedded in a record of
	type record of record {
	  integer a,
	  integer b
	} MyListType

	template MyListType mw_myBaseListTemplate := { ?, { a := 1, b := 2 }, ?, { a := 3, b := 4 } }

	template MyListType mw_myListTemplate1 modifies mw_myBaseListTemplate := { [1] := { a := 42 } }
	//Content of field ″a″ of the second element is modified,
	//the content of mw_myListTemplate1 is: { ?, { a := 42, b := 2 }, ?, { a := 3, b := 4 } }

	template MyListType mw_myListTemplate2 modifies mw_myBaseListTemplate := { -, { a := 42 } ,- }
	//Content of field ″a″ of the second element is modified, and the 
	//record of is truncated after the third element: { ?, { a := 42, b := 2 }, ? }

EXAMPLE 4:	Modified in-line template
	// Given 
	template MyRecordType m_setup :=
	{
		field1 := 75,
		field2 := "abc",
		field3 := true
	}

	// Could be used to define an in-line modified template of Setup
	// pco1.send (modifies m_setup := {field1:= 76});

EXAMPLE 5:	Modified parameterized template
	// Given 
	template MyRecordType m_myTemplate1(integer p_myPar):=
	{
		field1 := p_myPar,
		field2 := "A string",
		field3 := true
	}

	// then a modification could be 
		template MyRecordType m_myTemplate2(integer p_myPar) modifies m_myRecTemplate1 :=
		// field1 is parameterized in m_myTemplate1 and remains also parameterized in m_myTemplate2
	{
		field2 := "A modified string"
	}

EXAMPLE 6:	Default values of modified parameterized templates
	// Given 
	template MyRecordType m_myTemplate11 (integer p_int := 5 ):=
	  // p_int has the default value 5
	{
		field1 := p_int,
		field2 := "A string",
		field3 := true
	}

	// then possible template modifications are 
	template MyRecordType m_myTemplate12(integer p_int) modifies m_myTemplate11 :=
	  // p_int had a default value in m_myTemplate11 but has none in this template
	{
		field2 := "B string"
	}

	template MyRecordType m_myTemplate13(integer p_int := 0) modifies m_myTemplate12 := { }
	  // p_int has the default value 0
	  // no change is made to the template's content, but only to the default value of p_int

	template MyRecordType m_myTemplate14(integer p_int := - ) modifies m_myTemplate13 :=
	  // p_int inherits the default value 0 from its parent m_myTemplate13
	{
	  field2 := "C string"
	}

	template MyRecordType m_myTemplate15(integer p_int := - ) modifies m_myTemplate14 :=
	  // p_int inherits the default value 0 from m_myTemplate13 via m_myTemplate14
	{
	  field2 := "D string"
	}

	template MyRecordType m_myTemplate16(integer p_int) modifies m_myTemplate15 := { }
	  // p_int has no default value; no change in the template's content


	template MyRecordType m_myTemplate17(integer p_int := - ) modifies m_myTemplate16 :=
	  // causes an error as p_int has no default value in the parent template m_myTemplate16
	{
	  field2 := "E string"
	}



27.7	Optional attributes
The optional attribute can be used to indicate that optional fields of constants, module parameters or templates of record and set types are implicitly set to omit.
Syntactical Structure
optional

Semantic Description
TTCN‑3 constants, module parameters, and templates can have an optional attribute. Also, TTCN-3 language elements that contain such definitions, i.e. module, group, function, altstep, test case, control, and component type definitions can have an optional attribute. When an optional attribute is associated to a function, altstep, test case, control or component type definitions, it shall have effect on all the constants, module parameters, and templates declared within these definitions and not on the enframing definition itself.
Special optional strings:
The following strings are the predefined (standardized) optional attributes.
a)	"implicit omit" means that all optional fields, that have no assigned value definition in the statement on which the attribute operates, are set to omit. This applies recursively to the optional fields of the entity and to subfields of the mandatory fields.
b)	"explicit omit" means that all optional fields, that have no assigned value definition in the statement on which the attribute operates, are left undefined. This applies recursively to the optional fields of the entity and to subfields of the mandatory fields.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	Data type, port type, procedure signature and variable definitions and import statements shall not have an optional attribute associated to them directly. When an optional attribute is associated to module, group, function, altstep, test case, control or component type containing such definitions, it shall not have any effect on the included data type, port type, procedure signature, variable or import statement.
Examples
type record MyRecord1 {
  integer a,
  boolean b optional
}
type record MyRecord2 {
  MyRecord1 m
}

// reference templates with explicitly set fields
template MyRecord1 mw_myTemplate1 := { a := ?, b := omit }
template MyRecord2 mw_myTemplate2 := { m := { a := ?, b := omit }}

// reference templates
template MyRecord1 mw_myTemplate1a := {a := ? } // b is undefined
template MyRecord1 mw_myTemplate1b := {a := ? } with {optional "explicit omit"} // b is undefined

template MyRecord2 mw_myTemplate2a := {} // m and its subfields are undefined
template MyRecord2 mw_myTemplate2b := { m := { a := ?}}; // m.b is undefined

// templates with attribute

template MyRecord1 mw_myTemplate11 := { a := ? } with {optional "implicit omit"}
  // same as mw_myTemplate1, b is set to omit


template MyRecord2 mw_myTemplate21 := { m := { a := ?}} with {optional "implicit omit"}
  // same as mw_myTemplate2, by recursive application of the attribute
template MyRecord2 mw_myTemplate22 := { m := mw_myTemplate1a } with {optional "implicit omit"}
  // same as mw_myTemplate2, by recursive application of the attribute

template MyRecord2 mw_myTemplate23 := {} with {optional "implicit omit"}
  // same as mw_myTemplate2a, m remains undefined

template MyRecord2 mw_myTemplate24 := { m := mw_myTemplate1b } with {optional "implicit omit"}
  // same as mw_myTemplate2b, the attribute on the lower scope is not overwritten
template MyRecord2 mw_myTemplate25 := { m := MyTemplate1b } 
  with {optional override "implicit omit"}
  // same as mw_myTemplate2, the attribute on the lower scope is overwritten

  // implicitly omitted fields stay omitted after assignment
template MyRecord1 mw_myTemplate3a := mw_myTemplate1a with {optional "implicit omit"}
  // same as mw_myTemplate1, b is set to omit
template MyRecord1 mw_myTemplate3b := mw_myTemplate3a;
  // same as mw_myTemplate1, b is set to omit, by implicit omit attribute of mw_myTemplate3a
template MyRecord1 mw_myTemplate3c := mw_myTemplate3a with {optional "explicit omit"}
  // same as mw_myTemplate1, b is set to omit, by implicit omit attribute of mw_myTemplate3a

